Example nº 2
Buckling analysis of a steel frame

CivilFEM Manual of Advanced Examples
Example nº 2 – Table of Contents

2. EXAMPLE Nº 2: BUCKLING ANALYSIS OF A STEEL FRAME

2.1 AIM

The aim of this exercise is to show the capabilities that ANSYS and CivilFEM offer for the first and second order buckling analysis of a structure. The sample structure will be a biembedded steel frame.

2.2 DESCRIPTION OF THE EXAMPLE

The example consists in analysing the buckling behaviour of a frame represented in the following figure, when a 200 kN vertical force F acts on it, according to Eurocode 3:



The steel used is S 275 (UNE code) and the hot rolled sections:

HE120	for the piers
IPE 200	for the lintel

2.3 RESULTS TO BE OBTAINED

According to Eurocode 3 two indirect methods for the calculation of the buckling load can be used:

- First order analysis with amplified sway movements. It is only applicable if $\frac{\delta}{h}\frac{V}{H} \leq 0,25$ (condition that, as it will be seen ahead, is not fulfilled in this case).
- First order analysis with sway-mode buckling lengths.

These indirect methods will provide, as it is seen more ahead, a load of buckling smaller than the one of design of 200 kN, reason why it will be necessary to perform a different analysis, also admitted by Eurocode 3, consisting in using a direct method: second order elastic analysis with global and local (member) imperfections. With this method the stresses will be studied in each one of the members of the structure.

2.4 CALCULATION LOG

2.4.1 Indirect method (first order buckling)

The procedure to follow will be the following one:

Select the Eurocode 3 for the steel code to use and the International System as the units system to use (default values).

Next the parameters needed to build the model (geometry) are defined. This way, it is easy to modify the model to adapt it to other geometries, or to test different solutions, to perform an optimization analysis, etc.

The first step in the analysis of the structure is to choose the materials. The steel described previously will be chosen; next the hot rolled sections for each cross section of the frame are chosen.

The member properties are defined: buckling length and coefficients (the calculation is detailed ahead, in the chapter 2.5.1).

The model will be built using by BEAM188 elements.

The next step is to assign to the section properties and type of element to the beams.

The model is created defining nodes and elements, considering that the attributes of the elements of the piers and lintel are different. It is important to take into account the orientation of the cross section in the beam to follow the specifications of the model previously described. At the lintel, the default orientation is the one required, but in the piers it is necessary to rotate the beams 90º so that the section is placed as desired. To do so, the element orientation node (third node) will be used.

Once the model is the analysis options are set as linear static structural.

The boundary conditions are placed in the base of each pillar and the force is applied at the knot described at the beginning of the example. The horizontal loading of 10 kN entered to verify the sway-mode relationship of the frame, does not influence the buckling checking results.

After loading the model, it is solved.

Finally the results are checked according to Eurocode 3 and the results are plotted.

This first stage of the calculation also serves to verify that the relationship $\frac{\delta}{h}\frac{V}{H} \leq 0{,}25$ is not fulfilled.

2.4.1.1 Preprocess

```
FINISH
~CFCLEAR,,1

! Geometric Parameters
h = 5.0
b = 10.0

/PREP7

! Material Properties
~CFMP,1,LIB,STEEL,UNE,S 275
! Sections
~SSECLIB,1,1,5,2   !HE 120 A
~SSECLIB,2,1,1,7   !IPE 200
! Member properties according to Eurocode 3
~MEMBPRO,1,EC3,ALL,h,1.0,1.0,1.0,0.0,1.0,1.0,1.0,1.0,0.8,0,1.07,1.0
! Element type
ET,1,BEAM188

! Defines a beam property set: section, element, offset, etc.
~BMSHPRO,1,BEAM,1,1,,,188,1,0
~BMSHPRO,2,BEAM,2,2,,,188,1,0

! Modelling
N,1,0,0
N,11,0,h
FILL,1,11
N,21,b,h
FILL,11,21
N,31,b,0
FILL,21,31
N,100,-1,0,0 ! Beam orientation node

! Properties activation for left pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
E,1,2
EGEN,10,1,1
EMODIF,ALL,3,100
```

```
! Properties activation for lintel elements
MAT,1   $ TYPE,1  $  SECNUM,2
E,11,12
EGEN,10,1,11
! Properties activation for right pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
ESEL,NONE
E,21,22
EGEN,10,1,21
EMODIF,ALL,3,100
ESEL,ALL
```

2.4.1.2 Solution

```
/SOLU
! Static Structural Analysis
ANTYPE,STATIC
! Boundary Conditions
D,1,ALL
D,31,ALL
! Applied Load
F,11,FY,-200000
F,11,FX,  10000
! Solve
SOLVE
```

2.4.1.3 Postprocess

```
! Checking according to Eurocode n° 3
/POST1
! Number of the alternative to be read
~CFSET,,1
! Compression Buckling Checking
~CHKSTL,BUCK_CMP, ,-Z
! Plot Results
~PLCSEC3,BEAM,1,I
~PLLSSTL,NCR_Y,1
~PLLSSTL,CRT_TOT,1
```

2.4.2 Direct method (second order analysis)

The preprocessing of this second phase will be made ion the same way as for the indirect method, except for the generation of the model because a second order analysis will be performed, in which the imperfections are introduced directly in the geometry.

This is why the definition of the location of nodes differs from the previous method. In this case each node will have by abscissa the sum of the sway-mode displacement $x_1 = \frac{y}{20}$ and the deformed shape, approximated by $x_2 = e_{0,d}\frac{27}{4}(5-y)\frac{y^2}{5^3}$, with the maximum value of $e_{0,d}$ given by: $e_{0,d}$ = 1,83 cm, according to Fig.5.5.1 of Eurocode 3 (see Appendix B)

In the following figure both deformations are shown.



Another way to define the initial deformed one would consist in performing an eigenvalues buckling analysis, obtaining not only Euler's critical load, but also the deformed shape, that can be scaled with the factor $e_{0,d}$ (see Appendix A). In this example the analytical deformation stated before has been introduced directly to simplify the process.

Large deflections will be considered; therefore the analysis will be static and non-linear. In order to solve it, the iterative procedure of Newton-Raphson will be used with a minimum of 15 substeps for each loadstep, to achieve at the convergence.

Next, the boundary conditions are applied, the load in the specified node is applied and th model is solved.

Finally the resulting stresses are obtained to verify that the allowed maximum stress is not exceeded.

2.4.2.1 Preprocess

```
FINISH
~CFCLEAR,,1

! Geometric Parameters
h   = 5.0
b   = 10.0

/PREP7
! Material Properties
~CFMP,1,LIB,STEEL,UNE,S 275
! Sections
```

```
~SSECLIB,1,1,5,2   !HE 120 A
~SSECLIB,2,1,1,7   !IPE 200
! Element Type
ET,1,BEAM188

! Defines a beam property set: section, element, offset, etc.
~BMSHPRO,1,BEAM,1,1,,,188,1,0
~BMSHPRO,2,BEAM,2,2,,,188,1,0

! Modelling
N, 1,0.000,0
N, 2,0.004,h/10
N, 3,0.009,2*h/10
N, 4,0.015,3*h/10
N, 5,0.022,4*h/10
N, 6,0.028,5*h/10
N, 7,0.033,6*h/10
N, 8,0.036,7*h/10
N, 9,0.036,8*h/10
N,10,0.033,9*h/10
N,11,0.025,h
N,21,b+0.025,h
FILL,11,21
N,31,b,0
FILL,21,31
N,100,-1,0,0 ! Beam orientation node

! Properties activation for left pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
E,1,2
EGEN,10,1,1
EMODIF,ALL,3,100
! Properties activation for lintel elements
MAT,1  $   TYPE,1 $  SECNUM,2
E,11,12
EGEN,10,1,11
! Properties activation for riqht pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
ESEL,NONE
E,21,22
EGEN,10,1,21
EMODIF,ALL,3,100
ESEL,ALL
```

2.4.2.2 Solution

```
/SOLU
! Static Structural Analysis
ANTYPE,STATIC
! Non-linear analysis
OUTRES,ALL,ALL
LNSRCH,ON
NLGEOM,ON ! Large Displacements
NSUBST,15 ! Minimum number of loadsteps
! Boundary conditions
D,1,ALL
D,31,ALL
D,ALL,UZ
! Applied load
F,11,FY, -200000
! Solve
SOLVE
FINISH
```

2.4.2.3 Postprocess

```
/POST1
! Number of the alternative to be read
SET,1
! Plot Nodal Solution
/ESHAPE,1     ! Plots section shape
PLNSOL,S,X    ! X-Component Stress
```

2.5 RESULTS

2.5.1 Indirect method (first order analysis)

2.5.1.1 Buckling checking at section level



2.5.1.2 Compression buckling checking

The following figure shows the buckling resistance of the compressed member (120 kN):



Buckling calculation strength of a compressed member

The following figure represents graphically the obtained results using Eurocode 3, by means of a relation that shows the proximity to the limits established by the code. An equal to one criterion shows that the design forces and moments are equal to the code limit forces and moments. A criterion greater than one shows that the code is not fulfilled and a criterion smaller than one shows that the norm is fulfilled (the structure complies the code criteria). In this case it is observed that the loaded pier does not fulfil the code in any of its points:



Compression buckling checking criterion

The obtained criterion for the pier is greater than one. This shows that it has a load (200 kN) acting on it greater than the maximum allowed one (120 kN), according to Eurocode 3.

The condition $\frac{\delta}{h}\frac{V}{H} \leq 0{,}25$ is not fulfilled, because

$$\frac{\delta}{h}\frac{V}{H} = \frac{0.1194}{5}\frac{200000}{10000} = 0.4776 > 0.25$$

where δ is the horizontal displacement at node 11 of the sway-mode frame.

The distribution coefficients are calculated as follows:

- $\eta_1 = 0$ (embedding)
- $\eta_2 = \frac{K_c}{K_c + K_{11}}$ where K_C is the stiffness coefficient of the pier and K_{11} the stiffness coefficient of the beam:

$$K_C = \frac{I_C}{l_C} = \frac{231}{500} = 0.462\text{cm}^3 \qquad K_{11} = \frac{I_b}{l_b} = \frac{1946}{1000} = 1.94\ \text{cm}^3$$

According to figure E.2.2. of Eurocode 3 (see Appendix C) a β = 1.070 ratio is obtained. If the Spanish code EA-95 was used it would give a β = 0.743 value because Eurocode 3 does not contemplate the fact that one of the piers is unloaded.

If using the EA-95 code it was assumed that both piers are loaded the same, $\beta = 0.743\sqrt{2}$ = 1.050 would be obtained. This is a value similar to the one provided by Eurocode 3.

2.5.2 Direct method (second order analysis)

2.5.2.1 Calculation of normal stresses



It is verified that by this procedure the structure complies with the code:

Maximum obtained stress	212 MPa
Maximum allowed stress	275 MPa

2.6 SUMMARY

With this example the reader has learned to solve different types of analyses (linear and non-linear) as well as to perform buckling code checks.

APPENDIX A. Calculation of buckling eigenvalues (log)

```
FINISH
~CFCLEAR,,1

! Geometric Parameters
h = 5.0
b = 10.0

/PREP7
! Material Properties
~CFMP,1,LIB,STEEL,UNE,S 275
! Sections
~SSECLIB,1,1,5,2   !HE 120 A
~SSECLIB,2,1,1,7   !IPE 200
! Member properties according to Eurocode 3
~MEMBPRO,1,EC3,ALL,h,1.0,1.0,1.0,0.0,1.0,1.0,1.0,1.0,0.8,0,1.07,1.0
! Element type
ET,1,BEAM188

! Defines a beam property set: section, element, offset, etc.
~BMSHPRO,1,BEAM,1,1,,,188,1,0
~BMSHPRO,2,BEAM,2,2,,,188,1,0

! Modelling
N,1,0,0
N,11,0,h
FILL,1,11
N,21,b,h
FILL,11,21
N,31,b,0
FILL,21,31
N,100,-1,0,0 ! Beam orientation node

! Properties activation for left pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
E,1,2
EGEN,10,1,1
EMODIF,ALL,3,100
! Properties activation for lintel elements
MAT,1   $ TYPE,1  $  SECNUM,2
E,11,12
EGEN,10,1,11
! Properties activation for right pier elements
MAT,1  $  TYPE,1  $  SECNUM,1
ESEL,NONE
E,21,22
EGEN,10,1,21
EMODIF,ALL,3,100
ESEL,ALL

/SOLU
! Static Structural Analysis
ANTYPE,STATIC
! Store the stresses for the buckling analysis
PSTRES,ON
! Boundary Conditions
D,1,ALL
D,31,ALL
! Applied Load
F,11,FY,-1  ! Unit Load
! Solve
SOLVE
FINISH

/SOLU
```

```
! Buckling analysis (eigenvalues)
ANTYPE,BUCKLE
BUCOPT,LANB,1  ! Block Lanczos, extract 1 mode
MXPAND,1,,,YES ! Expand the mode
! Resolución
SOLVE
FINISH
```

APPENDIX B. Eurocode 3 – Calculation values for the equivalent initial curvature $e_{0,d}$



Cross-section		Method of global analysis	
Method used to verify resistance	Section type and axis	Elastic or Rigid - Plastic or Elastic - Perfectly plastic	Elasto-plastic (plastic zone method)
Elastic [5.4.8.2]	Any	$\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{el}/A$	-
Linear plastic [5.4.8.1(12)]	Any	$\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{pl}/A$	-
Non-linear plastic [5.4.8.1(1) to (11)]	I-section yy-axis	$1,33\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{pl}/A$	$\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{pl}/A$
	I-section zz-axis	$2,0\, k_\gamma e_{eff}/\epsilon$	$k_\gamma e_{eff}/\epsilon$
	Rectangular hollow section	$1,33\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{pl}/A$	$\alpha(\bar{\lambda} - 0,2)\, k_\gamma W_{pl}/A$
	Circular hollow section	$1,5\, k_\gamma e_{eff}/\epsilon$	$k_\gamma e_{eff}/\epsilon$

$k_\gamma = (1 - k_\delta) + 2 k_\delta \bar{\lambda}$ but $k_\gamma \geq 1,0$

Buckling curve	α	e_{eff}	k_δ			
			$\gamma_{M1} = 1,05$	$\gamma_{M1} = 1,10$	$\gamma_{M1} = 1,15$	$\gamma_{M1} = 1,20$
a	0,21	$\ell/600$	0,12	0,23	0,33	0,42
b	0,34	$\ell/380$	0,08	0,15	0,22	0,28
c	0,49	$\ell/270$	0,06	0,11	0,16	0,20
d	0,76	$\ell/180$	0,04	0,08	0,11	0,14

Non-uniform members:

Use value of W_{el}/A or W_{pl}/A at centre of buckling length ℓ

Figure 5.5.1 Design values of equivalent initial bow imperfection $e_{o,d}$

APPENDIX C. l/L Buckling length ratio of a pier for a sway-mode frame



Fig E.2.2 = I/L Buckling length ratio for a sway-mode frame